```
> # Solve for y1(x), y2(x) by the method of Frobenius,
> # using maple assist for substitution and solving the
  recurrences.
> # Problem. Solve x^2 y'' - x(2+x) y' + 2x y = 0
> # Answers: y1 = x^3(1+x/4+x^2/20+x^3/120+x^4/840+x^5/6720 + ...)
> #          y2 = x^0(12 +12x + 6x^2 + x^4/2 + x^5/10 + ...)
> de:=x^2*diff(y(x),x,x)-x*(2+x)*diff(y(x),x)+2*x*y(x)=0;
> dsolve({de},y(x),series);
```

$$de := x^2 \left(\frac{d^2}{dx^2}\, y(x)\right) - x\,(2+x)\left(\frac{d}{dx}\, y(x)\right) + 2\,x\,y(x) = 0$$

$$y(x) = \_C1\, x^3 \left(1 + \frac{1}{4}\,x + \frac{1}{20}\,x^2 + \frac{1}{120}\,x^3 + \frac{1}{840}\,x^4 + \frac{1}{6720}\,x^5 + O(x^6)\right) + \_C2 \left(12 + 12\,x + 6\,x^2 + 2\,x^3 + \frac{1}{2}\,x^4 + \frac{1}{10}\,x^5 + O(x^6)\right) \quad (1)$$

```
> z1:=sum(a(n)*x^(n+3),n=0..infinity); # Frobenius trial solution,
  a(0)=1
> subs(y(x)=z1,de);
> expand(%);
> simplify(lhs(%));
```

$$z1 := \sum_{n=0}^{\infty} a(n)\, x^{n+3}$$

$$x^2 \left(\frac{\partial^2}{\partial x^2}\left(\sum_{n=0}^{\infty} a(n)\, x^{n+3}\right)\right) - x\,(2+x)\left(\frac{\partial}{\partial x}\left(\sum_{n=0}^{\infty} a(n)\, x^{n+3}\right)\right) + 2\,x\left(\sum_{n=0}^{\infty} a(n)\, x^{n+3}\right) = 0$$

$$\sum_{n=0}^{\infty}\left(a(n)\, x^n x^3 n^2 + 5\, a(n)\, x^n x^3 n + 6\, a(n)\, x^n x^3\right) - 2\left(\sum_{n=0}^{\infty}\left(a(n)\, x^n x^3 n + 3\, a(n)\, x^n x^3\right)\right) - \left(\sum_{n=0}^{\infty}\left(a(n)\, x^n x^3 n + 3\, a(n)\, x^n x^3\right)\right) x + 2\,x\left(\sum_{n=0}^{\infty} a(n)\, x^n x^3\right) = 0$$

$$\sum_{n=0}^{\infty} a(n)\, x^{n+3}\left(n^2 + 5\,n + 6\right) - 2\left(\sum_{n=0}^{\infty} a(n)\, x^{n+3}\,(n+3)\right) - \left(\sum_{n=0}^{\infty} a(n)\, x^{n+3}\,(n+3)\right) x + 2\,x\left(\sum_{n=0}^{\infty} a(n)\, x^{n+3}\right) \quad (2)$$

```
> Q1:=n->a(n)*(n^2+5*n+6-2*n-6); # Coeff of x^(n+3)
> Q2:=n->a(n)*(-n-3+2); # Coeff of x^(n+4)
> Q1(0); # Coeff of x^3 should be zero (its the indicial equation)
> Q1(n)+Q2(n-1)=0; # recursion relation for sequence a[n], valid
  for n>=1
```

$$Q1 := n \rightarrow a(n)\left(n^2 + 3\,n\right)$$

$$Q2 := n \rightarrow a(n)\,(-n-1)$$

$$0$$

$$a(n)\left(n^2+3n\right)-a(n-1)n=0 \qquad (3)$$

```
> # Use maple rsolve() to solve the first recurrence relation
> eq:=Q1(n)+Q2(n-1)=0;
> ic:=a(0)=1;
> aa:=unapply(rsolve({eq,ic},a),n);
> seq(aa(i),i=0..10);
```

$$eq := a(n)\left(n^2+3n\right)-a(n-1)n=0$$

$$ic := a(0)=1$$

$$aa := n \to \frac{6}{\Gamma(n+4)}$$

$$1, \frac{1}{4}, \frac{1}{20}, \frac{1}{120}, \frac{1}{840}, \frac{1}{6720}, \frac{1}{60480}, \frac{1}{604800}, \frac{1}{6652800}, \frac{1}{79833600}, \frac{1}{1037836800} \qquad (4)$$

```
> z2:=0*ln(x)*y1(x)+sum(b(n)*x^(n+0),n=0..infinity); # Change 0 to
  k if it fails
> subs(y(x)=z2,de);
> expand(%);
```

$$z2 := \sum_{n=0}^{\infty} b(n)\,x^n$$

$$x^2\left(\frac{\partial^2}{\partial x^2}\left(\sum_{n=0}^{\infty} b(n)\,x^n\right)\right)-x(2+x)\left(\frac{\partial}{\partial x}\left(\sum_{n=0}^{\infty} b(n)\,x^n\right)\right)+2x\left(\sum_{n=0}^{\infty} b(n)\,x^n\right)=0$$

$$\sum_{n=0}^{\infty}\left(b(n)\,x^n n^2-b(n)\,x^n n\right)-2\left(\sum_{n=0}^{\infty} b(n)\,x^n n\right)-\left(\sum_{n=0}^{\infty} b(n)\,x^n n\right)x+2x\left(\sum_{n=0}^{\infty} b(n)\,x^n\right)=0 \qquad (5)$$

```
> simplify(lhs(%));
> R1:=n->(n*(n-1)-2*n)*b(n); # Coeff of x^n
> R2:=n->(-n+2)*b(n); # Coeff of x^(n+1)
```

$$\sum_{n=0}^{\infty} b(n)\,x^n n\,(n-1)-2\left(\sum_{n=0}^{\infty} b(n)\,x^n n\right)-\left(\sum_{n=0}^{\infty} b(n)\,x^n n\right)x+2x\left(\sum_{n=0}^{\infty} b(n)\,x^n\right)$$

$$R1 := n \to (n(n-1)-2n)\,b(n)$$

$$R2 := n \to (-n+2)\,b(n) \qquad (6)$$

```
> # Use maple rsolve() to solve the second recurrence relation
> eq:=R1(n)+R2(n-1)=0;
> ic:=b(0)=12; # To match the dsolve output. Normally choose b(0)=1
> bb:=unapply(rsolve({eq,ic},b),n);
> seq(bb(n),n=0..10);
```

$$eq := (n(n-1)-2n)\,b(n)+(-n+3)\,b(n-1)=0$$

$$ic := b(0)=12$$

$$bb := n \to \frac{12}{\Gamma(n+1)}$$

$$12, 12, 6, 2, \frac{1}{2}, \frac{1}{10}, \frac{1}{60}, \frac{1}{420}, \frac{1}{3360}, \frac{1}{30240}, \frac{1}{302400} \qquad (7)$$